

J. D. HEGSETH	91	WORD UNJUMBLE PROGRAM
J. D. HEGSETH	92	BUG RACE
J. D. HEGSETH	93	MASTER PROGRAM CATALOG
RANDOLPH T. WEDDING	96	LEVEL I LINEAR REGRESSION
RANDOLPH T. WEDDING	97	LEVEL II LINEAR REGRESSION
TRS-80 USERS GROUP	99	CHECK (PRINTS CHECKS)
TRS-80 USERS GROUP	100	BILLS (DATA FILES)
TRS-80 USERS GROUP	102	CHECK YEAR TO DATE DATA

\*\*\*\*\* THE NEWSLETTER IS STILL IN TROUBLE \*\*\*\*\*

WE NEED PROGRAMS AND ARTICLES OF ANY KIND. WE ARE NOW MORE THAN 90 DAYS BEHIND WITH THE NEWSLETTER. PLEASE HELP US GET UP TO DATE. SEND IN ANY PROGRAMS OR ARTICLES YOU HAVE WE WILL BE GLAD TO PRINT THEM IN THE NEWSLETTER. IF YOU COULD SEND ANY PROGRAMS ON TAPE IT WILL MAKE THEM EASIER FOR US TO PUT THEM IN THE NEWSLETTER. WE WILL MAKE A HARD COPY THE RIGHT SIZE FORM YOUR TAPE. YOUR EDITOR: R. GORDON LLOYD

DEAR MR. LLOYD,

I RECEIVED YOUR NOTE THAT YOU WERE SHORT OF MATERIAL FOR THE NEWSLETTER. ENCLOSED IS A TAPE CONTAINING FIVE MISCELLANEOUS PROGRAMS THAT MIGHT BE SUITABLE.

MOST ARE SELF-EXPLANATORY BUT A FEW NOTES ARE IN ORDER. THEY ARE ALL ORIGINAL PROGRAMS ALTHOUGH MOST WERE INSPIRED BY ARTICLES OR IDEAS IN OTHER PUBLICATIONS.

WORD UNSCRAMBLING PROGRAM: INSPIRED BY A DISCUSSION IN THE ORANGE COUNTY USERS GROUP NEWSLETTER.

BUG RACE: FROM TURTLE RACE BY RON MOSHLIS IN MAY 79 OF THIS NEWSLETTER.

CATALOG: ENTIRELY ORIGINAL BUT PROBABLY SIMILAR TO OTHER CATALOG OR MENU PROGRAMS.

CATALOG: LINES 290 & 300 SPECIFY RANGE OF DISK NUMBERS. (I CURRENTLY USE 101 TO 122 AND 998 TO 999) CHANGE FOR YOUR DISK NUMBERS.

LINE 340 SET D=HIGHEST DISK # IN USE.

LINE 500 FOR SINGLE DISK, CHANGE DIRVE #1 TO #0

LINE 615 DELETE (FOR A SPECIAL SYSTEM PROGRAM DISK)

LINE 620 FOR SINGLE DISK, DELETE :1 AFTER MENU.

P. S. DELETE DATA LINES AND REPLACE WITH YOUR OWN USING THE FORMAT OF DISK #, # OF PROGRAMS, FREE SPACE, PROGRAM TITLES. EACH INDIVIDUAL DISK CONTAINS A SIMILAR BUT SMALLER PROGRAM CALLED MENU WHICH ONLY CONTAINS DATA FOR THAT DISK.

CORDIALLY,  
JACK D. HEGSETH  
BOX 62, 1921 E. 208TH ST.  
SPANAWAY, WA 98387

```

10 REM WORD UNJUMBLE PROGRAM
20 REM BY J. D. HEGSETH  COPYRIGHT APRIL 1979
100 CLS:A$=""
110 PRINT"THIS IS A WORD UNSCRAMBLING PROGRAM. "
120 PRINT"ENTER ANY SEQUENCE OF LETTERS UP TO NINE. "
130 PRINT"THE PROGRAM WILL PRINT OUT ALL COMBINATIONS. "
140 PRINT:INPUT A$
150 A=LEN(A$):IF A<1 OR A>9 THEN PRINT"1 TO 9 LETTERS ONLY":
    FOR I=1 TO 1000:NEXT I:GOTO 100
160 PRINT:INPUT"DO YOU WANT LINEPRINTER OUTPUT (0=NO 1=YES)";B
200 REM DISASSEMBLE WORD
210 F=1
220 FOR I=1 TO A
230 A$(I)=MID$(A$,I,1)
240 F=F*I
250 NEXT I
260 PRINT
270 ON A GOSUB 1000,2000,3000,4000,5000,6000,7000,8000,9000
280 PRINT:PRINT"THESE ARE THE";F;
    " COMBINATIONS OF THE ORIGINAL";A;" LETTERS. "
290 IF B=1 THEN LPRINT:LPRINT"THESE ARE THE";F;
    " COMBINATIONS OF THE ORIGINAL";A;" LETTERS. "
300 REM END ROUTINE OR REPEAT ROUTINE
310 PRINT:INPUT"ANOTHER WORD (YES OR NO)";B$
320 B$=LEFT$(B$,1)
330 IF B$="Y" THEN RUN ELSE IF B$="N" THEN END ELSE GOTO 310
500 REM PRINT ROUTINE
510 WD$=""
520 FOR I0=1 TO A
530 WD$=WD$+A$(I0)
540 NEXT I0
550 PRINT WD$,:IF B=1 THEN LPRINT WD$,
560 RETURN
1000 REM 1 LETTER SORT
1010 GOSUB 500
1020 RETURN
2000 REM 2 LETTER SORT
2010 FOR I=1 TO 2
2020 GOSUB 500
2030 A$(0)=A$(1)
2040 A$(1)=A$(2)
2050 A$(2)=A$(0)
2060 NEXT I
2070 RETURN
3000 REM 3 LETTER SORT
3010 FOR J=1 TO 3
3020 GOSUB 2000
3030 FOR I=0 TO 2
3040 A$(I)=A$(I+1)
3050 NEXT I
3060 A$(I)=A$(0)
3070 NEXT J
3080 RETURN
4000 REM 4 LETTER SORT
4010 FOR K=1 TO 4
4020 GOSUB 3000
4030 FOR I=0 TO 3
4040 A$(I)=A$(I+1)
4050 NEXT I

```

```

4060 A$(I)=A$(0)
4070 NEXT K
4080 RETURN
5000 REM 5 LETTER SORT
5010 FOR L=1 TO 5
5020 GOSUB 4000
5030 FOR I=0 TO 4
5040 A$(I)=A$(I+1)
5050 NEXT I
5060 A$(I)=A$(0)
5070 NEXT L
5080 RETURN
6000 REM 6 LETTER SORT
6010 FOR M=1 TO 6
6020 GOSUB 5000
6030 FOR I=0 TO 5
6040 A$(I)=A$(I+1)
6050 NEXT I
6060 A$(I)=A$(0)
6070 NEXT M
6080 RETURN
7000 REM 7 LETTER SORT
7010 FOR N=1 TO 7
7020 GOSUB 6000
7030 FOR I=0 TO 6

```

```

A 100/$ 140/$ 150/3 150/$ 220 230/$ 230($ 270 280 290
520 530($ 2030($2 2040($2 2050($2 3040($2 3060($2
4040($2 4060($2 5040($2 5060($2 6040($2 6060($2 7040($2
7060($2 8040($2 8060($2 9040($2 9060($2
B 160 290 310/$ 320/$2 330/$2 550
F 210 240/2 280 290
I 150/2 220 230/2 240 250 2010 2060 3030 3040/2 3050
3060 4030 4040/2 4050 4060 5030 5040/2 5050 5060
6030 6040/2 6050 6060 7030 7040/2 7050 7060 8030
8040/2 8050 8060 9030 9040/2 9050 9060
IO 520 530 540
J 3010 3070
K 4010 4070
L 5010 5070
M 6010 6070
N 7010 7070
O 8010 8070
P 9010 9070
WD 510/$ 530/$2 550/$2

```

### THE BITPICKER'S TOOLBOX

By Steve MacGregor, 3701 W Wethersfield, Phoenix, Arizona 85029

If you have a 16K (or more) Level II computer, and are sending a program to someone with only 4K of RAM, a kind thing to do is to check to see that it will load and run on his machine. The easy way to do this is to answer the MEMORY SIZE? question with 20481 to simulate a 4K machine. Any BASIC program that will run then will run in 4K. Note that the number to type is the address of the last byte of RAM, plus two, in decimal. And, of course, you can simulate a 16K RAM on a machine with 32K or more, etc.

```

7040 A$(I)=A$(I+1)
7050 NEXT I
7060 A$(I)=A$(0)
7070 NEXT N
7080 RETURN
8000 REM 8 LETTER SORT
8010 FOR O=1 TO 8
8020 GOSUB 7000
8030 FOR I=0 TO 7
8040 A$(I)=A$(I+1)
8050 NEXT I
8060 A$(I)=A$(0)
8070 NEXT O
8080 RETURN
9000 REM 9 LETTER SORT
9010 FOR P=1 TO 9
9020 GOSUB 8000
9030 FOR I=0 TO 8
9040 A$(I)=A$(I+1)
9050 NEXT I
9060 A$(I)=A$(0)
9070 NEXT P
9080 RETURN
9999 END

```

```

10 REM BUG RACE BY JD HEGSETH APRIL 1979
70 REM DEFINE BUG STRINGS
80 A$=STRING$(3,174)+CHR$(140)
90 B$=STRING$(4,128)
100 REM LAYOUT COURSE
110 CLS
120 PRINT@153,"B U G R A C E";:PRINT@209,
" S I L V E R F I S H D O W N S";
130 FOR I=15616 TO 15679:POKE I,143:NEXT I
140 FOR I=15744 TO 15807:POKE I,140:NEXT I
150 FOR I=15872 TO 15935:POKE I,140:NEXT I
160 FOR I=16000 TO 16063:POKE I,188:NEXT I
170 FOR I=13 TO 31:SET(126,I):SET(127,I):NEXT I
180 W1=0:W2=0:W3=0
200 REM PRINT BUGS
210 B1=320:B2=448:B3=576
220 PRINT@B1,"1";A$;:PRINT@B2,"2";A$;:PRINT@B3,"3";A$;
230 B1=B1+1:B2=B2+1:B3=B3+1
240 PRINT@782,"O N Y O U R M A R K":FOR I=1 TO 250:NEXT I
250 PRINT@846,"G E T S E T":FOR I=1 TO RND(250)+100:NEXT I
260 PRINT@910,"G O O O O O O O !"
300 REM START RACE
310 RANDOM
320 T=0
330 I=RND(3):T=T+1
340 ON I GOTO 400,500,600
400 REM MOVE BUG 1
410 PRINT@B1,B$;
420 B1=B1+1
430 PRINT@B1,A$;
440 IF B1=379 THEN 700
450 GOTO 330
500 REM MOVE BUG 2
510 PRINT@B2,B$;
520 B2=B2+1
530 PRINT@B2,A$;
540 IF B2=507 THEN 700
550 GOTO 330
600 REM MOVE BUG 3
610 PRINT@B3,B$;
620 B3=B3+1
630 PRINT@B3,A$;
640 IF B3=635 THEN 700
650 GOTO 330
700 REM WINNER
710 PRINT@782,"T H E W I N N E R I S B U G # ";I
730 GOSUB 800:PRINT@846,"W I T H A L E A D O F ";L;" M M";
740 T=T/20:PRINT@910,"A N D A T I M E O F ";T;" S E C";
750 FOR I=1 TO 2000:NEXT I
760 REM RETURN TO STARTING LINE
770 FOR I=B1 TO 322 STEP -1:GOSUB 900:NEXT I:
FOR I=B2 TO 450 STEP -1:GOSUB 900:NEXT I:
FOR I=B3 TO 578 STEP -1:GOSUB 900:NEXT I
780 PRINT@782,CHR$(30):PRINT@910,CHR$(30)
790 GOTO 210
800 REM FIND WINNING LEAD
810 ON I GOTO 820,830,840
820 L1=507-B2:L2=635-B3:W1=W1+1:GOTO 850
830 L1=379-B1:L2=635-B3:W2=W2+1:GOTO 850

```

```

840 L1=379-B1:L2=507-B2:W3=W3+1
850 IF L1>L2 THEN L=L2 ELSE L=L1
860 L=L*3
870 PRINT@0,"BUG NUMBER :", " 1", " 2", " 3";
880 PRINT@64,"TOTAL WINS :",W1,W2,W3;
890 RETURN
900 REM MOVE BUGS LEFT
910 PRINT@I,B$;:PRINT@I-1,A$;
920 RETURN
999 END

```

```

A 80/$ 220/$3 430/$ 530/$ 630/$ 910/$
B 90/$ 410/$ 510/$ 610/$ 910/$
B1 210 220 230/2 410 420/2 430 440 770 830 840
B2 210 220 230/2 510 520/2 530 540 770 820 840

B3 210 220 230/2 610 620/2 630 640 770 820 830
I 130/3 140/3 150/3 160/3 170/4 240/2 250/2 330 340
  710 750/2 770/6 810 910/2
L 730 850/2 860/2
L1 820 830 840 850/2
L2 820 830 840 850/2
T 320 330/2 740/3
W1 180 820/2 880
W2 180 830/2 880
W3 180 840/2 880

```

```

BUG NUMBER :      1          2          3
TOTAL WINS  :      2          2          3

```

BUG RACE  
SILVERFISH DOWNS

```

1.....
2.....
3.....

```

THE WINNER IS BUG # 2  
WITH A LEAD OF 3 MM  
AND A TIME OF 8.25 SEC

TRS-80 QUALITY SOFTWARE

LEVEL I. AND LEVEL II.		LEVEL II	
#1. IDI-I CASSETTE DATA BASE	\$20.	#11. WORD-I WORD PROCESSOR	\$25.
#2. INV-I INVENTORY CONTROL	\$20.	#15. MAIL-I NAME AND ADDRESS	\$25.
#3. STOCK-I SECURITY INFO.	\$10.	#16. SORT-I SORT UTILITY	\$10.
#4. BANK-I CHECK BALANCE	\$10.	#17. STAT-I STATISTICS	\$10.
#5. FINANCE-I STOCK-I & BANK-I	\$15.	#18. KEY-I KEY-ACCESS	\$10.
		#19. SALE-I SALE ANALYSIS	\$10.
		#20. UTIL-I SORT-I & KEY-I	\$16.

DISKETTE.  
#12. MAIL-III MAILING LIST \$35  
#14. WORD-III WORD PROCESSOR \$35  
#21. INV-III INVENTORY CONTROL \$35  
#22. KEY-III KEY RANDOM ACCESS \$15.

MICRO ARCHITECT  
96 DOTHAN ST.  
ARLINGTON, MA 02174

PRICE INCLUDES POSTAGE, CASSETTER & DOCU.

```

10 REM MASTER PROGRAM CATALOG
20 REM J.D. HEGSETH OCTOBER 1979
30 REM FILENAME "CATALOG"
100 REM INSTRUCTIONS AND INITIALIZATION
110 CLS:CMD"R"
120 PRINTTAB(19);"M A S T E R C A T A L O G"
130 PRINT:PRINT:PRINT
    "DISK MENUS MAY BE DISPLAYED INDIVIDUALLY OR SEQUENTIALLY."
140 PRINT:PRINT"DO YOU WANT HARD COPY (Y OR N) ? ";
150 X$=INKEY$
160 X$=INKEY$:IF X$="N" THEN LP=0:GOTO 200
170 IF X$<"Y" THEN 160
180 IF PEEK(14312)>127 THEN PRINT"LINE PRINTER NOT READY!";
    GOTO 140 "
190 LP=1
200 PRINT X$
210 PRINT:PRINT"NEXT DISK:      > N <"
220 PRINT:PRINT"SPECIFIC DISK:  > S <"
230 PRINT:PRINT"DISK (N OR S) ? ";
240 X$=INKEY$
250 X$=INKEY$:IF X$="N" THEN 320
260 IF X$<"S" THEN 250
270 CLS
280 DK=0:PRINT"WHICH DISK # ";:INPUT DK
290 IF DK>100 AND DK<123 THEN 320
300 IF DK=998 OR DK=999 THEN 320
310 PRINT:PRINT
    "I N V A L I D   D I S K   # !   R E - E N T E R !";
    GOTO 280
320 REM DATA READOUT
330 RESTORE
340 IF D=999 THEN 630
350 READ D,P,F:REM D= DISK #, P= # OF PROGRAMS, F= FREE SPACE
360 IF DK=0 OR DK=D THEN 380
370 FOR J=1 TO P:READ A$:NEXT J:GOTO 340
380 CLS:PRINT
    "FILE DIRECTORY FOR DISK #";D,"FREE =";F;"GRANS":PRINT
390 IF LP=0 THEN 410
400 LPRINT STRING$(3,138):LPRINT"FILE DIRECTORY FOR DISK #";D,
    "FREE =";F;"GRANS":LPRINT CHR$(138)
410 FOR J=1 TO P:READ A$
420 PRINT J;A$;
430 IF INT(J/2)*2<J THEN PRINTTAB(32);ELSE PRINT
440 IF LP=0 THEN 470
450 LPRINT J;A$;
460 IF INT(J/2)*2<J THEN LPRINTTAB(32);ELSE LPRINT CHR$(138)
470 NEXT J
480 PRINT:IF INT(J/2)*2=J THEN PRINT
485 IF LP=1 THEN IF INT(J/2)*2=J THEN LPRINT CHR$(138)
490 PRINT
    "NEXT > N <, SPECIFIC > S <, RERUN CATALOG > R <, PROGRAM > P <";
500 X$=INKEY$
510 X$=INKEY$
520 IF X$="N" THEN DK=0:GOTO 340
530 IF X$="S" THEN 270
540 IF X$="R" THEN RUN
550 IF X$<"P" THEN 510
560 REM SELECT A PROGRAM FROM DISPLAYED MENU
570 CLS:PRINT"TO RUN A SELECTED PROGRAM:"

```

```

580 PRINT:PRINT"1 INSERT DISK #";D;"INTO DRIVE # > 1 <"
590 PRINT"2 PRESS > M < WHEN READY"
600 X$=INKEY$
610 X$=INKEY$:IF X$<"M" THEN 610
620 LOAD"MENU:1",R
630 CLS:PRINT"THAT WAS THE LAST DISK IN THE CATALOG"
640 PRINT:PRINT"PRESS > ENTER < TO RERUN THE CATALOG"
650 INPUT AN$
660 RUN
670 END
1000 REM DATA (DISK#), # PGMS, FREE, PGM NAMES)
1010 DATA101,9,3,INVENTORY,FILE,SPACE WAR,GLOBAL BEARINGS,
PILLBOX 2,QSL GRAPHICS,STAR TREK,SPACE TREK II,
TROLLS GOLD,MUSIC
1020 DATA102,16,1,BLACK JACK,BACKGAMMON,ACEY DEUCY,MATCH 23,
KING,HAMURABI,TIC TAC TOE,RANDOM WORD,L2 RANDOM WORD,
SEARCH,LUNAR LANDER,SHOOTING STARS,MINITREK,MIND READER,
ROACH RACE,ENTRAPMENT
1030 DATA103,7,2,GRAPHIC STAR TREK,STAR TREK 3,STAR TREK 1,
JUMBLE 2,SEARCH 1,STAR TREK 2,UNSCRAMBLE
1040 DATA104,18,1,DOOMSDAY,JUMBLE 1,MEMORY LETTER TEST,
STINGRAY,RUSSIAN ROULETTE,WHEEL OF FORTUNE,
TOWERS OF BABEL,DECISION,MEMORY NUMBER TEST,FLIPPER,
STARS,ODD ONE,SPYSHIP,HORSE,SCISSORS,CRAPPS,STAR BLAZER,
TIGER SHARK
1050 DATA105,21,1,FRONT COVER,WEIRD,RAT RACE,RANDOM AD,
FIRESIDE,LEFT RIGHT,BLOCKS,HERRING,LAUNCH,BLINKER,SNOOPY,
SNOW,STEP AD 1,STEP AD 2,BASE CONVERSION DRILL,SPEEDY,
LIFE,COUNT,ROAD RACE,FIFTEEN,SANTA PARAVIA
1060 DATA106,15,4,BARTENDER,CALORY COUNTER,METRIC CONVERSION,
UNIVERSAL CALENDAR,BASE CONVERSION,CALCULATOR,
VACATION LIST,TELECODE,MESSAGE CENTER,NIGHT CHECK LIST,
EXPENSE ACCOUNT,BABY SITTER,DRUNKOMETER,REMEMBER,MILEAGE
1070 DATA107,5,2,FINANCE PACKAGE 1,FINANCE PACKAGE 2,
FINANCE PACKAGE 3,GENERAL KNOWLEDGE DRILL,
TOUCH TYPING DRILL
1080 DATA108,2,26,GENERAL PURPOSE FILE
PROGRAM AND DATA FILES,MAIL LIST
1090 DATA109,3,29,TEXT-80 TEXT EDITOR,
COMPUTRONICS WORD PROCESSOR,EDITOR AND SCREEN PRINTER
1100 DATA110,7,1,ERODE,AROUND THE HORN,CODE PRACTICE,
CODE PRAC. (EXPANDED),CRIBBAGE,ROBOT CHASE,SAFARI
1110 DATA111,19,3,SPEED GRAPHICS,RANDOM GRAPHICS,
TRS-80 DISPLAY 1,TRS-80 DISPLAY 2,TRS-80 DISPLAY 3,
BIG GRAPHICS,BUG DRAW,BUG RACE,BIG GRAPHICS (CONDENSED),
STRING GRAPHIC ROUTINE,GRAPHIC ROUTINES (ML),GRAPHIC,
GRAPHIC 2,SKETCH
1115 DATAHEINRICH LIFE,SOUND EFFECT ROUTINE,SOUND DEMO PROGRAM,
RANDOM ART,MAZE
1120 DATA112,5,1,DOG STAR,AWARI,NO TRUMP BRIDGE,DRAW POKER,
STUD POKER
1130 DATA113,4,19,LUNAR LANDER,KEYBOARD 80,INVASION FORCE,
CHECKERS 80
1140 DATA114,7,2,SPACE TREK 4,SUPER SUB,RAMROM PATROL,
KLINGON CAPTURE,TIE FIGHTER,BALLOON RACE,COMPUTER NIM
1150 DATA115,4,8,TREASURE HUNT,PIRATES COVE,VOODOO CASTLE,
THE COUNT
1160 DATA116,11,2,BASE CONVERSIONS,DUPE QSO FILE,GYPSY,
UNJUMBLE,XMAS LIST,ADDITION TEST,SUBTRACTION TEST,
MULTIPLY/DIVIDE TEST,FRACTION DECIMAL TEST,

```

```

QUEST ADVENTURE,POPULATION SIMULATION
1170 DATA117,8,15,MISSION IMPOSSIBLE,TINY PILOT,TAPEDISK,DCV,
EDITOR/ASSEMBLER,RENUMBER 16K,RENUMBER 32K,TBUG
1180 DATA118,8,1,TANK RUN,RACE CAR,YAHTZEE,SHOOTING GALLERY,
TREASURE DUNGEON,BRIDGE ENGINEER,BATTLESHIP,ROBOT HUNT
1190 DATA119,5,1,WESTWARD 1847,NUCLEAR MELTDOWN,ISOLATE,
EVEREST EXPEDITION,GREED 2
1200 DATA120,6,1,SPACE TARGET,GRAPHIC TREK 2000,INVASION WORG,
SAUCERS,STAR WARS,LUNAR LANDER
1210 DATA121,6,1,EGYPTIAN PHAROAH,TANGLE GRAPH,WORD SCRAMBLE,
PSYCHIC FORTUNE,CHECKERS,DRAW POKER
1220 DATA122,1,28,OIL BARON
9980 DATA998,6,32,MASTER CATALOG,TAPEDISK/CMD,DCV/CMD,
BASIC DIRECTORY,INKEY ROUTINE,40 CHAR LLISTER
9990 DATA999,13,5,MASTER CATALOG,TEST1/CMD,TEST2/BAS,
GETTAPE/BAS,GETDISK/BAS,TAPEDISK/CMD,DCV/CMD,DISKDUMP/BAS,
DUMPDISK/BAS,DASSEMBLP/BAS,TIMEDATE/TXT,HEXPOKE/BAS,
REFSERCH/TXT

```

```

A 370/$ 410/$ 420/$ 450/$
AN 650/$
D 340 350 360 380 400 500
DK 280/2 290/2 300/2 360/2 520
F 350 380 400
J 370/2 410 420 430/2 450 460/2 470 480/2 485/2
LP 160 190 390 440 485
P 350 370 410
R 620
X 150/$ 160/$2 170/$ 200/$ 240/$ 250/$2 260/$ 500/$
510/$ 520/$ 530/$ 540/$ 550/$ 600/$ 610/$2

```

#### MASTER CATALOG

DISK MENUS MAY BE DISPLAYED INDIVIDUALLY OR SEQUENTIALLY.

DO YOU WANT HARD COPY (Y OR N) ? N

NEXT DISK: > N <

SPECIFIC DISK: > S <

DISK (N OR S) ?

FILE DIRECTORY FOR DISK # 101 FREE = 3 GRAMS

1 INVENTORY FILE	2 SPACE WAR
3 GLOBAL BEARINGS	4 PILLBOX 2
5 QSL GRAPHICS	6 STAR TREK
7 SPACE TREK II	8 TROLLS GOLD
9 MUSIC	

NEXT > N <, SPECIFIC > S <, RERUN CATALOG > R <, PROGRAM > P <

TO RUN A SELECTED PROGRAM:

```

1 INSERT DISK # 101 INTO DRIVE # > 1 <
2 PRESS > M < WHEN READY

```

TRS-80' DUST COVERS

- \* PROTECT YOUR VALUABLE INVESTMENT FROM DUST AND DIRT BUILD UP
- \* SAVE \$ ON MAINTENANCE, DOWNTIME AND THEY LOOK GREAT
- \* IMPROVE SYSTEM RELIABILITY AND HELP PREVENT KEY BOUNCE
- \* MADE FROM BLACK CLOTH-BACKED VINYL NOT SEE-THROUGH PLASTIC
- \* COVERS HAVE REINFORCED CUT-OUTS FOR PLUGS AND CABLES

TAPE	\$3.99	
KEYBOARD	\$4.99	COMPUTER PRODUCTS
VIDEO	\$7.99	BOX 1475
	-----	MINOT, N. D. 58701
	\$16.97	

COMPLETE SET OF THREE COVERS \*\*\* \$14.95 \*\*\*  
 LONG VIDEO (EXPLAN. INTER.) ADD \$2.00  
 ALL PRICES POSTPAID N. D. RESIDENTS ADD 3% TAX  
 'TRS-80 IS A TRADEMARK OF TANDY CORP. AND RADIO SHACK

2502 4TH ST N.W. #7  
 MINOT, N. D. 58701  
 OCT 23, 1979

DEAR MR. LLOYD,

I RECENTLY HAD THE FREE VOLUME SENSITIVITY MODIFICATION DONE TO MY UNIT AND WAS CONCERNED WHEN IT CAME BACK WITH A \$14 BILL. AFTER READING THE WRITE UP ON THE BILL I FOUND THAT ALL THEY HAD DONE WAS TO CLEAN THE INTERIOR. AFTER BEING BITTEN ONCE, I SAID NEVER AGAIN AND PROCEEDED TO DESIGN AND MAKE DUST COVERS FOR ALL 3 PIECES OF MY SYSTEM. AFTER MAKING A FEW SETS OF COVERS FOR FRIENDS IN MY LOCAL AREA I WONDERED HOW MANY OTHER COMPUTER OWNERS HAVE OR WILL HAVE TO PAY THIS CLEANING FEE. I AM NOW OFFERING THESE COVERS TO ONE AND ALL, AT A MODEST PRICE. THE AD APPEARS ELSEWHERE IN THIS ISSUE. INCIDENTLY, THE SENSITIVITY MODIFICATION WORKS FINE.

SINCERELY,  
 WILLIAM J. CAREY

COMPREHENSIVE RADIOLOGY BILLING PGM FOR TRS-80

HANDLES UNLIMITED NO OF ACCOUNTS  
 SOFTWARE OPERATES WITH 48K MEM 2 DRIVES LINE PRINTER  
 AGES, CYLCLE BILLS, ZIP CODE SORTS, PRINTS INS FORMS  
 SEARCH FILES BY NAME OR AC#

ITEMIZED STATEMENT, INS FORM, CORRECT OR ADD TO ACCOUNT ANYTIME  
 ALPHA SORT PRINTOUT FOR CROSSFILE  
 SIMPLE 3 LETTER PROCEEDURE ENTRY (IVP, UGI, ETC)  
 BILLING ANAYISIS ON SCREEN OR PRINTOUT  
 CAN BE ADAPTED TO OTHER SPECIALTIES

PROGRAM IN USE ONE YEAR AND FULLY DEBUGGED

PROGRAM FOR \$595.00\*  
 P. J. PALMISANO MD (214) 592-2133)

4 November 1979

Mr. R. G. Lloyd  
 TRS-80 Users Group  
 7554 Southgate Rd.  
 Fayetteville, N.C. 28304

Dear Mr. Lloyd:

Enclosed are copies of two programs for line fitting by linear regression. The first of these is in Level I BASIC and will run in 16K as written, or by reducing the number of data pairs to a more reasonable level can run in 8K. The program also includes a plotting routine for the CRT but since a hard copy can only be obtained with a screen printer this is more a curiosity than a genuine advantage.

The second program is an adaptation of the first to Level II or Disk BASIC and includes data enter or save on disk rather than tape as well as a line printer plotting routine. Both programs have a correction routine for faulty data entry.

In addition I have a number of other statistical programs including second order parabolic regression, third order parabolic regression, multiple linear regression for 2 X values, multiple linear regression for 3 X values, randomized block analysis of variance, split plot analysis of variance, hyperbolic regression (for fitting enzyme kinetic data) and a generalized non-linear least squares fit using the Marquardt algorithm adaptable to fitting any equation with up to 12 parameters. All of the above are in Level II and a few are also available in Level I.

If any readers are interested I can supply any of the above on cassette with a program listing, documentation and references for \$10 each, post paid. A diskette with all of the listed programs can be supplied for \$50. A number of specialized programs for biochemical data reduction and analysis are also available and I would be glad to correspond with anyone concerning them.

Sincerely yours,

*R. T. Wedding*  
 Randolph T. Wedding

5445 Brittany Avenue  
 Riverside, CA 92506

```

10 REM LEVEL I LINEAR REGRESSION WITH SCREEN PLOT
20 REM REQUIRES 16 K FOR 100 X,Y PAIRS. CAN BE REDUCED TO 8 K
30 REM FOR SMALLER DATA SETS. INCLUDES DATA LOAD AND SAVE ON
40 REM CASSETTE AND DATA CORRECTION ROUTINE
50 /
60 / RANDOLPH T. WEDDING
70 / 5445 BRITTANY AVE.
80 / RIVERSIDE, CA 92506
80 CLEAR 100
90 DIM A(300)
100 CLS
110 PRINT TAB(25)"LINEAR REGRESSION"
120 S=0:T=0:U=0:V=0:W=0
130 FOR I=1 TO 25:READ A(I+200):NEXT I
140 INPUT "ENTER EXPERIMENT AND DATE";A$,B$
150 INPUT "NUMBER OF DATA PAIRS";N
160 INPUT "DATA FROM TAPE, NO=0, YES = 1, TAPE READY";Z
170 IF Z=0 THEN 200
180 FOR I=1 TO N:INPUT#-1,A(I),A(100+I):NEXT I
190 GOTO 230
200 FOR I=1 TO N:PRINT "VALUE OF X, & Y("I)"
210 INPUT A(I),A(100+I)
220 NEXT I
230 INPUT "ANY ERRORS (1=YES, 2=NO)";V:IF V=2 THEN 260
240 INPUT "DATA PAIR NUMBER WITH ERROR";I
250 PRINT "Y( ";I," ) X( ";I," )":INPUT A(I),A(100+I):GOTO 230
260 INPUT "RECORD DATA ON TAPE 0=NO, 1=YES, RECORDER READY";Z
270 IF Z=0 THEN 290
280 FOR I=1 TO N:PRINT#-1,A(I),A(100+I):NEXT I
290 FOR I=1 TO N:S=S+A(I):T=T+A(I+100):U=U+(A(I)*A(I))
300 V=V+(A(I+100)*A(I+100)):W=W+(A(I)*A(I+100)):NEXT I
310 E=U-(S*S)/N:F=V-(T*T)/N:G=W-(S*T)/N
320 H=G/E:K=T/N:L=S/N:M=K-(H*L)
330 O=H*G:P=F-O:Q=O/(P/(N-2)):R=O/F
340 X=R:GOSUB 960 :R=Y
350 B=P/(N-2):B=SQR(B)
360 X=N:GOSUB 960 :L=Y:C=B/L
370 S=P/(N-2):S=S/E:X=S:GOSUB 960 :S=Y
380 Z=(P/(N-2))/B
390 D=H/S
400 CLS
410 PRINT TAB(20) "RESULTS OF LINEAR REGRESSION"
420 PRINT
430 PRINT A$;TAB(50) B$
440 PRINT
450 PRINT "INTERCEPT =",M, "SLOPE =",H
460 PRINT
470 PRINT "F VALUE FOR REGRESSION =",Q
480 PRINT "CORRELATION COEFFICIENT, R =",R;TAB(40) "FOR N =",N
490 PRINT "RESIDUAL STANDARD DEVIATION =",B
500 PRINT "STD. ERROR OF RANDOM SAMPLE =",C
510 PRINT "STD. ERR. OF SLOPE =",S," , OF INTERCEPT =",Z
520 PRINT "T VALUE FOR SLOPE =",D," , FOR INTERCEPT =",M/Z
530 I=(200+N-2)
540 U=100*C*A(I)/(T/N)
550 PRINT "T REQUIRED AT 5% =",A(I)

```

```

560 PRINT "SAMPLING ERROR % =",U
570 INPUT "PRINT DATA = 1, PLOT = 2";A:ON A GOTO 880 ,630
580 DATA 12,706,4,303,3,182,2,776,2,571
590 DATA 2,447,2,365,2,306,2,262,2,228
600 DATA2,201,2,179,2,160,2,145,2,131
610 DATA 2,120,2,110,2,101,2,093,2,086
620 DATA 2,080,2,074,2,069,2,064,2,060
630 INPUT "ENTER X-MAX,X-MIN,X-SPACE";A,B,C
640 INPUT "ENTER Y-MAX,Y-MIN,Y-SPACE";D,E,F
650 G=(100/A-B)*C:K=(40/D-E)*F
660 CLS:PRINT A$;TAB(50) B$
670 FOR I=20 TO 120:SET(I,47):SET(I,7):NEXT I
680 FOR I=7 TO 47:SET(20,I):SET(120,I):NEXT I
690 FOR I=20+G TO 120:SET(I,46):I=I+G:NEXT I
700 FOR I=7+K TO 47:SET(21,I):I=I+K:NEXT I
710 J=100:FOR I=1 TO N:A(I)=((A(I)/(A-B))*100)+20:NEXT I
720 FOR J=101 TO 100+N:A(J)=((A(J)/(D-E))*40):NEXT J
730 FOR J=101 TO 100+N:A(J)=47-A(J):NEXT J
740 FOR I=1 TO N:SET(A(I),A(I+100)):NEXT I
750 GOSUB 830
760 INPUT "2";Z:IF Z=2 THEN 770
770 M=47-((40/D-E)*M)
780 FOR I=1 TO 100:L=20+I
790 O=(I*.4)*(H*(A/D))
800 SET(L,M-O):IF M-O<8 THEN 830
810 NEXT I
820 END
830 Q3=47-((M/(D-E))*40)
840 RESET(20,Q3)
850 Q4=47-((M+(H*A))/(D-E))*40)
860 RESET(120,Q4)
870 RETURN
880 Q=1:R=13:IF N<13 THEN R=N
890 CLS:PRINT A$;TAB(50) B$
900 PRINT "NO ", "X", "Y", "Y(CALC)"
910 FOR I=Q TO R:PRINT I,A(I),A(I+100),A(I)+H*M
920 NEXT I
930 INPUT "MORE, 1=YES, 2=NO";F:ON F GOTO 940 ,570
940 Q=R+1:R=Q+12:IF R>N THEN R=N:CLS:GOTO910
950 STOP
960 IF X=0 THEN Y = 0:RETURN
970 IF X>0 THEN 990
980 PRINT "ROOT OF NEGATIVE NUMBER":STOP
990 Y=X*.5:Z=0
1000 W=(X/Y-Y)*.5
1010 IF (W=0) + (W=Z) THEN RETURN
1020 Y=Y+W:Z=W:GOTO 1000

```

```

10 REM LINEAR REGRESSION WITH SCREEN OR LINE PRINTER PLOT
20 REM WILL HANDLE 40 X,Y PAIRS IN 16K, BUT IS MORE
30 REM FLEXIBLE IN 32K. INCLUDES DATA LOAD AND SAVE
40 REM ON DISK AND DATA CORRECTION ROUTINE. PLOT ON
50 REM LINE PRINTER IS ONLY APPROXIMATE - X AXIS IS
60 REM ON LEFT EDGE OF PRINTOUT, Y AXIS ACROSS.
70 / RANDOLPH T. WEDDING
80 / 5445 BRITTANY AVE
90 / RIVERSIDE, CA 92506
100 CLEAR 400
110 DIM X(40),Y1(40),A(25),X1(40),Y2(40),Y3(40)
120 CLS
130 PRINT TAB(25)"LINEAR REGRESSION"
140 S=0:T=0:U=0:V=0:W=0
150 FOR I=1TO25:READ A(I):NEXT I
160 INPUT "ENTER EXPERIMENT, AND DATE";A$,B$
170 INPUT "NUMBER OF DATA PAIRS";N
180 INPUT "ENTER DATA ID";Z$
190 INPUT "DATA FROM DISK, NO=0, YES = 1";Z6
200 IF Z6=0 THEN 270
210 OPEN "I",1,Z$
220 FOR I=1TON
230 INPUT#L,X(I),Y1(I)
240 NEXT I
250 CLOSE 1
260 GOTO 400
270 FOR I=1TON:PRINT "VALUE OF X, & Y("I")";
280 INPUT X(I),Y1(I)
290 NEXT I
300 INPUT "ANY ERRORS (1=YES,2=NO)";V2:IF V2=2 THEN 330
310 INPUT "DATA PAIR NUMBER WITH ERROR";I
320 PRINT "Y(";I;"): X(";I;"):INPUT A(I),A(100+I):GOTO 300
330 INPUT "RECORD DATA ON DISK 0=NO, 1=YES";Z7
340 IF Z7=0 THEN 400
350 OPEN "O",1,Z$
360 FOR I=1TON
370 PRINT#L,X(I),Y1(I)
380 NEXT I
390 CLOSE 1
400 FOR I=1TON:S=S+X(I):T=T+Y1(I):U=U+X(I)*I
410 V=V+Y1(I)[2:W=W+(X(I)*Y1(I)):NEXT I
420 E=U-(S*S)/N:F=V-(T*T)/N:G=W-(S*T)/N
430 H=G/E:K=T/N:L=S/N:M=K-(H*L)
440 O=H*G:P=F-O:Q=O/(P/(N-2)):R=O/F
450 R=SQR(R)
460 B=P/(N-2):B=SQR(B)
470 L=SQR(N):C=B/L
480 S=P/(N-2):S=S/E:S=SQR(S)
490 S1=(P/(N-2))/B
500 D=H/S
510 INPUT "READY FOR PRINTOUT";Z4
520 LPRINT TAB(20)."RESULTS OF LINEAR REGRESSION"
530 LPRINT " "
540 LPRINT A$;TAB(50) B$
550 LPRINT " "

```

```

560 LPRINT "INTERCEPT =",M, "SLOPE =",H
570 LPRINT " "
580 LPRINT "F VALUE FOR REGRESSION =",Q;" DATA ID = ";Z$
590 LPRINT "CORRELATION COEFFICIENT, R = ";R:TAB(40) "FOR N = ";N
600 LPRINT "RESIDUAL STANDARD DEVIATION =",B
610 LPRINT "STD. ERROR OF RANDOM SAMPLE = ";C
620 LPRINT "STD. ERR. OF SLOPE =",S;" , OF INTERCEPT = ";S1
630 LPRINT "T VALUE FOR SLOPE =",D;" , FOR INTERCEPT = ";M/S1
640 I=N-2
650 U=100*C+A(I)/(T/N)
660 LPRINT "T REQUIRED AT 5% =",A(N-1)
670 LPRINT "SAMPLING ERROR % =",U
680 LPRINT " "
690 INPUT "PRINT OUT DATA = 1, PLOT ON SCREEN = 2, PLOT ON PRINTER = 3";A9
700 DATA 12.706,4.303,3.182,2.776,2.571
710 DATA 2.447,2.365,2.306,2.262,2.228 :ON A9 GOTO 1000,750,1100
720 DATA2,201,2.179,2.160,2.145,2.131
730 DATA 2.120,2.110,2.101,2.093,2.086
740 DATA 2.080,2.074,2.069,2.064,2.060
750 INPUT "ENTER X-MAX, X-MIN, X-SPACE";A,B,C
760 INPUT "ENTER Y-MAX, Y-MIN, Y-SPACE";D,E,F
770 G=(100/A-B)*C:K=(40/D-E)*F
780 CLS:PRINT A$;TAB(50) B$
790 FOR I=20 TO 120:SET(I,47):SET(I,7):NEXT I
800 FOR I=7 TO 47:SET(20,I):SET(120,I):NEXT I
810 FOR I=20+G TO 120:SET(I,46):I=I+G:NEXT I
820 FOR I=7+K TO 47:SET(21,I):I=I+K:NEXT I
830 FOR I=1TON:X1(I)=((X(I)/(A-B))+100)+20:NEXT I
840 FOR I=1TON:Y2(I)=(Y1(I)/(D-E))*40
850 Y2(I)=47-Y2(I):NEXT I
860 FOR I=1TON:SET(X1(I),Y2(I)):NEXT I
870 GOSUB 950
880 INPUT "2";Z:IF Z=2 THEN 890
890 M=47-((40/D-E)*M)
900 FOR I=1 TO 100:L=20+I
910 O=(I*.4)*(H*(A/D))
920 SET(L,M-O):IF M-O<8 THEN 950
930 NEXT I
940 END
950 Q3=47-((M/(D-E))*40)
960 RESET(20,Q3)
970 Q4=47-((H*(H+A))/(D-E)*40)
980 RESET(120,Q4)
990 RETURN
1000 LPRINT "NO. ", "X", "Y", "Y(CALC)"
1010 B1=B*A(N-1)
1020 FOR I=1TON
1030 B2=Y1(I)-(X(I)*H+M):B3=SQR(B2+B2)
1040 IF B3>B1 THEN 1060
1050 LPRINT I,X(I),Y1(I),X(I)*H+M:GOTO 1070
1060 LPRINT I,X(I),Y1(I),X(I)*H+M:TAB(50)"*"
1070 NEXT I
1080 LPRINT "SIGMA X 'T' = ",B1
1090 INPUT "PLOT ON SCREEN OR PRINTER, 1 OR 2";Z5:ON Z5 GOTO 750,1100
1100 REM PLOTTING ROUTINE FOR LINE PRINTER

```

```

1110 INPUT"ENTER SYMBOL FOR POINTS";Q$
1120 INPUT"ENTER LABELS FOR X AND Y AXES";S$,T$
1130 INPUT "ENTER X-MAX, X-MIN";A1,B1
1140 C1=A1-B1
1150 INPUT"ENTER Y-MAX, Y-MIN";D1,E1
1160 F1=D1-E1
1170 FOR I=1TON
1180 X1(I)=(X(I)/C1)*30:Y2(I)=(Y1(I)/F1)*90:NEXT I
1190 FOR I=1TO30 :Y3(I)=(I*(C1/30)+H+M)/F1*90:NEXT I
1200 LPRINT" "
1210 LPRINT TAB(30) "(" ,E1:"TO",D1," - ",T$;" )
1220 LPRINT STRING$(3," ");STRING$(93,"-");
1230 LPRINT CHR$(10)
1240 L1=1
1250 FOR K=1TO30
1260 IF K=INT(X1(L1)) THEN 1280
1270 LPRINT STRING$(3," ");"I";STRING$(Y3(K)," ");" ";STRING$(93-Y3(K)," ");"I":GOTO 1320
1280 IF Y3(K)<Y2(L1) THEN 1300
1290 LPRINT X(L1);STRING$(Y2(L1)," ");Q$;STRING$(Y3(K)-Y2(L1)," ");
" ";STRING$(93-Y3(K)," ");"I":GOTO1310
1300 LPRINT X(L1);STRING$(Y3(K)," ");" ";" ";STRING$(Y2(L1)-Y3(K)," ");" ";
1310 L1=L1+1
1320 NEXT K
1330 LPRINT STRING$(3," ");STRING$(93,"-");
1340 LPRINT" "
1350 LPRINT "(" ,B1:"TO",A1," - ",S$;" )"

```


*Micropute  
Software*

# ORGAN

PRESENTS

by: JAMES WILLIAMS

CAN BE PLAYED ON ANY TRS-80 4K LEVEL II AND UP



ANYONE WHO CAN PLAY "CHOPSTICKS" WILL ENJOY THIS NEW, CLEVER USE OF THE TRS-80

REAL TIME TWO MANUAL KEYBOARD WITH 3 OCTAVES PLUS  
 PLAY SINGLE NOTES OR CHORDS  
 ASTERISK APPEARS ON DISPLAY AS KEY IS DEPRESSED  
 CONTROL VOLUME (2 LEVELS)  
 CONTROL TREMOLO (ALTERNATION BETWEEN NOTES 10 SPEEDS)  
 CONTROL QUALITY OF SOUND (11 SELECTIONS)  
 RECORD YOUR PERFORMANCE DIRECTLY ON TAPE  
 REAL TIME LISTENING WITH AMPLIFIER OR TAPE MODIFICATION

COMPLETE OPERATING AND HOOKUP INSTRUCTIONS INCLUDED WITH EACH ORGAN PROGRAM

SEND \$14.95 TO: MICROPUTE SOFTWARE, P.O. BOX 1943  
 ROCKY MOUNT, N.C. 27801

NORTH CAROLINA RESIDENTS PLEASE ADD 4% SALES TAX

Dear Mr. Lloyd,

I would like to inform the readers of an error in my calendar program published in the July issue (Pg.69). The error is easily corrected by changing lines 1370 and 1400 to the following:

```

1370 IF M1 = 0 AND M2 = 0 THEN 1240
1400 PRINT : GOTO 1240 (or LPRINT)

```

My thanks to Mr. Fred W. Smith of Randleman, NC for bringing the error to my attention.

Sincerely yours,

*Zvi Dershowitz*  
 Zvi Dershowitz  
 18 Harav Berlin St.  
 Jerusalem, Israel



```

1000 REM * CHECK *
1010 REM * TRS-80 USERS GROUP - JAN 80 *
1020 CLEAR 2000: DIM X0$(27), X1$(12)
1030 CLS: GOSUB 5470
1040 PRINT: PRINT: PRINT TAB(23) "CHECK WRITER"
1050 PRINT: INPUT "      STARTING CHECK NUMBER "; CN
1060 PRINT: INPUT "DATE (MM, DD, YY) (EX. 01, 01, 80) "; D$: PRINT
1070 DATA ONE, TWO, THREE, FOUR, FIVE, SIX, SEVEN, EIGHT, NINE
1080 DATA TEN, ELEVEN, TWELVE, THIRTEEN, FOURTEEN, FIFTEEN, SIXTEEN
1090 DATA SEVENTEEN, EIGHTEEN, NINETEEN, TWENTY, THIRTY, FORTY
1100 DATA FIFTY, SIXTY, SEVENTY, EIGHTY, NINETY
1110 DATA JAN, FEB, MAR, APR, MAY, JUN, JUL, AUG, SEP, OCT, NOV, DEC
2000 OPEN "R", 1, "BILL": GOSUB 6000
2010 OPEN "R", 2, X1$(VAL(LEFT$(D$, 2)))
3000 PRINT "NAME OF PAYEE - "; F$
3010 IF VAL(AM$)=0 THEN INPUT "AMOUNT OF CHECK "; AM$:
      GOSUB 6000: GOTO 3030
3020 PRINT "AMOUNT OF CHECK "; USING "$#####.###"; VAL(AM$)
3030 INPUT "WOULD YOU LIKE TO PRINT THIS CHECK "; PC$
3040 PRINT: IF EOF(1) THEN RUN
3050 IF PC$="N" GOSUB 6000: GOTO 3000
3060 PRINT "PRINT CHECK #"; CN
4000 A=VAL(AM$): A9$=AM$
4010 D1=VAL(LEFT$(D$, 2))
4020 D2=VAL(MID$(D$, 4, 2))
4030 D3$=RIGHT$(D$, 2)
4040 S1=39: S2=60: S3=63: S4=14
4050 S8$="#####": S9$="#####.###"
5000 LPRINT TAB(S1>D2; X1$(D1); " 19"; D3$;
5010 LPRINT TAB(S2) USING S9$; VAL(AM$);
5020 LPRINT TAB(S3) USING S8$; CN
5030 LPRINTCHR$(138)
5040 LPRINT TAB(S4) F$
5050 LPRINTCHR$(138)
5060 IF VAL(AM$)<=0 THEN 5440
5070 N1=INT(VAL(A1$))
5080 IF N1>99 THEN 5440
5090 IF N1=0 THEN 5120
5100 GOSUB 5350
5110 LPRINT "THOUSAND ";
5120 N1=INT(VAL(A2$))
5130 IF N1=0 THEN 5160
5140 GOSUB 5350
5150 LPRINT "HUNDRED ";
5160 N1=INT(VAL(A3$))
5170 IF N1>0 THEN 5200
5180 IF A>=1 THEN 5210
5190 GOTO 5230
5200 GOSUB 5350
5210 PRINT: PRINT
5220 LPRINT "DOLLARS ";
5230 A1=VAL(A4$)
5240 LPRINT "AND";
5250 LPRINT INT(A1); "CENTS"
5260 PRINT
5270 LPRINTCHR$(138)
5280 LPRINT TAB(8) P0$
5290 LPRINT TAB(8) P1$
5300 LPRINT TAB(8) P2$
5310 LPRINT TAB(8) "ACCOUNT NUMBER - "; P3$

```

```

5320 FOR S7=1 TO 11: LPRINTCHR$(138): NEXTS7
5330 GOSUB 9000
5340 CN=CN+1: GOSUB 6000: GOTO 3000
5350 IF N1<21 THEN 5410
5360 LPRINT X0$((N1-20)/10+20); " ";
5370 A3=N1-INT(N1/10)*10
5380 IF A3=0 THEN 5430
5390 LPRINT "- ";
5400 N2=N1: N1=A3
5410 LPRINT X0$(N1); " ";
5420 IF A3=N1 THEN N1=N2
5430 RETURN
5440 PRINT "***** V O I D *****"
5450 GOTO 5270
5460 END
5470 FOR I=1 TO 27
5480 READ X0$(I)
5490 NEXT I
5500 FOR I=1 TO 12
5510 READ X1$(I)
5520 NEXT I
5530 RETURN
6000 LR=LR+1
6010 GOSUB 7000
6020 PR=INT((LR-1)/2)+1
6030 GET 1, PR
6040 F$=NA$: P0$=NA$
6050 P1$=AD$
6060 P2$=CT$+ST$+ZP$
6070 P3$=AC$
6080 MA=VAL(AM$): MA$=STR$(MA): AM$=MID$(MA$, 2)
6090 IF MID$(RIGHT$(AM$, 2), 1, 1)="." THEN AM$=AM$+"0"
6100 IF INT(VAL(AM$))=VAL(AM$) THEN AM$=AM$+"."00
6110 IF VAL(AM$)>999.99 THEN M1=1: M2=2: M3=3: M4=2: GOTO 7030
6120 IF VAL(AM$)>99.99 THEN M2=1: M3=2: M4=2: C2=1: C3=2: GOTO 7040
6130 IF VAL(AM$)>9.99 THEN M3=1: M4=2: C3=2: A1$="0": A2$="0":
      GOTO 7050
6140 IF VAL(AM$)>.99 THEN M3=1: M4=2: C3=1: A1$="0": A2$="0": GOTO 7050
6150 A=VAL(AM$): RETURN
7000 SR=LR-2*INT((LR-1)/2)-1
7010 FIELD 1, SR*127 AS H$, 7 AS DA$, 30 AS NA$, 35 AS AD$,
      20 AS CT$, 3 AS ST$, 5 AS ZP$, 20 AS AC$, 7 AS AM$
7020 RETURN
7030 A1$=LEFT$(AM$, M1): REM THOUSAND
7040 A2$=MID$(AM$, M2, C2): REM HUNDRED
7050 A3$=MID$(AM$, M3, C3): REM DOLLARS
7060 A4$=RIGHT$(AM$, M4): REM CENTS
7070 RETURN
8000 PR%=INT((KEY%-1)/4)+1
8010 SR%=KEY%-4*(PR%-1)
8020 FIELD 2, ((SR%-1)*50) AS H1$, 5 AS U1$, 7 AS U2$, 8 AS U3$,
      30 AS U4$
8030 GET 2, PR%
8040 RETURN
9000 KEY%=KEY%+1
9010 GOSUB 8000
9020 CN$=STR$(CN): LSET U1$=MID$(CN$, 2): LSET U2$=A9$
9030 LSET U3$=D$: LSET U4$=NA$
9040 PUT 2, PR%
9050 RETURN

```

```

1000 REM * BILLS *
1010 REM * TRS-80 USERS GROUP - JAN 80 *
1020 CLS:PRINT
1030 X=0:N=0
1040 CLEAR 1000:CLS:CLOSE
1050 S$="BILL":PRINT:PRINT
2000 PRINT:PRINT "TYPE          1 - WRITE A FILE"
2010 PRINT:PRINT "                2 - READ A FILE"
2020 PRINT:PRINT "                3 - EDIT A FILE"
2030 PRINT:PRINT "                4 - SEARCH FOR A FILE"
2040 PRINT:PRINT "                5 - END ";N
2050 IF N=5 THEN CLOSE:END
2060 OPEN"R",1,S$
2070 CLS
2080 ON N GOSUB 3000,4010,5000,6000
2090 GOTO 1030
3000 GOSUB 9030:LR=LR+1
3010 GOSUB 9100
3020 PRINT@593,"NEXT RECORD NUMBER ";CHR$(94);LR;S$;" ";CHR$(93)
3030 PRINT@0,"DATE - TYPE 'END' TO STOP " TAB(30);
3040 LINEINPUTA$:J1=LEN(A$):LSETDA$=A$:IFLEN(A$)>7THEN3020
3050 IF A$="END" THEN RETURN
3060 PRINT"NAME"TAB(30);:LINEINPUTA$:J2=LEN(A$):LSETNA$=A$:
IFLEN(A$)>30THEN3060
3070 PRINT"ADD"TAB(30);:LINEINPUTA$:J3=LEN(A$):LSETAD$=A$:
IFLEN(A$)>35THEN3070
3080 PRINT"CITY"TAB(30);:LINEINPUTA$:J4=LEN(A$):LSETCT$=A$:
IFLEN(A$)>20THEN3080
3090 PRINT"STATE"TAB(30);:LINEINPUTA$:J5=LEN(S$):LSETST$=A$:
IFLEN(A$)>2THEN3090
3100 PRINT"ZIP"TAB(30);:LINEINPUTA$:J6=LEN(A$):LSETZP$=A$:
IFLEN(A$)>5THEN3100
3110 PRINT"ACCOUNT # "TAB(30);:LINEINPUTA$:J7=LEN(A$):LSETAC$=A$:
IFLEN(A$)>20THEN3110
3120 PRINT"AMOUNT"TAB(30);:LINEINPUTA$:J8=LEN(A$):LSETAM$=A$:
IFLEN(A$)>7THEN3120
3130 INPUT "IS THIS INFORMATION CORRECT (Y/N) ";X$
3140 IF X$="N" GOSUB 10000:GOTO 3020
3150 IF X$="Y" GOSUB 10000:GOSUB 11000:GOTO 4000
3160 GOTO 3130
4000 PUT 1,PR:CLS:GOTO 4020
4010 PRINT@192,
, "INPUT THE LR # FOR THE FILE YOU ARE WORKING ON ";:
INPUTLR:GOSUB9100:GOSUB11000
4020 PRINT@320,
" INPUT LABLES (L) OR ROLODEX (R) OR END (E) ";:
INPUT YY$
4030 IF YY$="E" THEN 1030
4040 IF YY$="R" THEN L=1:GOTO 4060
4050 PRINT:INPUT " INPUT THE # OF LABLES YOU NEED ";L
4060 GOSUB 9100
4070 ZP=VAL(ZP$)
4080 IF ZP=0 THEN RETURN
4090 GOSUB 11000
4100 FOR I=1 TO L
4110 IF S$="BILL" THEN GOSUB 14000:GOTO 4130
4120 GOSUB 12000
4130 NEXT I
4140 IF N=1 THEN CLS:GOTO 3000

```

```

4150 GOTO 1030
5000 CLS:PRINT@192,
"INPUT THE LR # FOR THE FILE YOU ARE WORKING ON ";:INPUTLR
5010 CLS:GOSUB 5310 :X=1
5020 PRINT@5,
" THERE ARE 8 LINES OF DATA WHICH WOULD YOU LIKE TO CHANGE"
5030 PRINT "DATE", "NAME", "ADD", "CITY"
5040 PRINT "STATE", "ZIP", "ACC (#)", "AMOUNT"
5050 PRINT "TYPE 'END' TO RETURN TO MENU ";:LINEINPUT CH$
5060 IF CH$="DATE" THEN 5160
5070 IF CH$="NAME" THEN 5170
5080 IF CH$="ADD" THEN 5180
5090 IF CH$="CITY" THEN 5190
5100 IF CH$="STATE" THEN 5200
5110 IF CH$="ZIP" THEN 5210
5120 IF CH$="ACC" THEN 5220
5130 IF CH$="AMOUNT" THEN 5230
5140 IF CH$="END" THEN 1040
5150 GOTO 5050
5160 LINEINPUT "NEW DATE ";A1$:GOTO 5250
5170 LINEINPUT "NEW NAME ";A2$:GOTO 5250
5180 LINEINPUT "NEW ADD ";A3$:GOTO 5250
5190 LINEINPUT "NEW CITY ";A4$:GOTO 5250
5200 LINEINPUT "NEW STATE ";A5$:GOTO 5250
5210 LINEINPUT "NEW ZIP ";A6$:GOTO 5250
5220 LINEINPUT "NEW ACCOUNT # ";A7$:GOTO 5250
5230 LINEINPUT "NEW AMOUNT ";A8$:GOTO 5250
5240 GOTO 5050
5250 LSET DA$=A1$:LSET NA$=A2$:LSET AD$=A3$:LSET CT$=A4$
5260 LSET ST$=A5$:LSET ZP$=A6$:LSET AC$=A7$:LSET AM$=A8$
5270 INPUT "ANY MORE CHANGES TO THIS FILE (Y/N) ";YN$
5280 CLS
5290 IF YN$="Y" GOSUB 11000:GOTO 5020
5300 PUT 1,PR
5310 GOSUB 9100
5320 ZP=VAL(ZP$):IF ZP=0 THEN RETURN
5330 GOSUB 11000
5340 PRINT
5350 A1$=DA$:A2$=NA$:A3$=AD$:A4$=CT$:A5$=ST$:A6$=ZP$:A7$=AC$:A8$=AM$
5360 IF X=1 THEN 5380
5370 RETURN
5380 GOTO 4020
6000 A=0:B=0:C=0:D=0:E=0:H=0
6010 LR=0:N=3
6020 CLS
6030 LR=LR+1
6040 GOSUB 9100
6050 ZP=VAL(ZP$):IF ZP=0 THEN CLOSE:RETURN
6060 GOTO 8010
7000 CLS
7010 GOSUB 11000:PRINT
7020 PRINT "DO YOU NEED A HARD COPY (Y/N) "
7030 PRINT " TO MAKE A CHANGE (C) "
7040 INPUT " TO STOP (END) ";U$
7050 IF U$="Y" GOSUB 12000
7060 IF U$="C" THEN CLS:GOTO 5010
7070 IF U$="END" THEN 1030
8000 CLS:GOTO 6030
8010 IF (A)OR(B)OR(C)OR(D)OR(E)=1 THEN 8110
8020 PRINT,"DATE", "NAME", "CITY"

```